```
> Demand := P = 60 - 4*Q/10;
```

$$Demand := P = 60 - \frac{2}{5} Q \quad (1)$$

```
> Revenue := R = P*Q;
```

$$Revenue := R = P\,Q \quad (2)$$

```
> eval(subs(Demand,Revenue));
```

$$R = \left(60 - \frac{2}{5} Q\right) Q \quad (3)$$

```
> MarginalRevenue := MR = diff(rhs(%),Q);
```

$$MarginalRevenue := MR = -\frac{4}{5} Q + 60 \quad (4)$$

```
> Production := Q = sqrt(10*E-5000);
```

$$Production := Q = \sqrt{10\,E - 5000} \quad (5)$$

```
> MarginalProduct := MPL = diff(rhs(Production),E);
```

$$MarginalProduct := MPL = \frac{5}{\sqrt{10\,E - 5000}} \quad (6)$$

```
> MarginalRevenueProduct := MRPL = MR*MPL;
```

$$MarginalRevenueProduct := MRPL = MR\,MPL \quad (7)$$

```
> LaborSupply := w = 2;
```

$$LaborSupply := w = 2 \quad (8)$$

```
> Profit := pi = R - w*E;
```

$$Profit := \pi = R - w\,E \quad (9)$$

```
> eval(subs(Revenue,Demand,Production,LaborSupply,Profit));
```

$$\pi = \left(60 - \frac{2}{5}\sqrt{10\,E - 5000}\right)\sqrt{10\,E - 5000} - 2\,E \quad (10)$$

```
> plot(rhs(%),E=600..1000,title="Profit");
```




```
> ProfitMax := MRPL=w;
```

$$ProfitMax := MRPL = w \quad (11)$$

```
> eval(subs(MarginalRevenueProduct,LaborSupply,%));
```

$$MR\,MPL = 2 \quad (12)$$

```
> eval(subs(MarginalRevenue,%));
```

$$\left(-\frac{4}{5}\,Q + 60\right) MPL = 2 \quad (13)$$

```
> eval(subs(Production,%));
```

$$\left(-\frac{4}{5}\,\sqrt{10\,E - 5000} + 60\right) MPL = 2 \quad (14)$$

```
> eval(subs(MarginalProduct,%));
```

$$\frac{5\left(-\frac{4}{5}\,\sqrt{10\,E - 5000} + 60\right)}{\sqrt{10\,E - 5000}} = 2 \quad (15)$$

```
> solve(%,{E});
```

$$\{E = 750\} \quad (16)$$

```
> simplify(eval(subs(%,Production)));
```

$$Q = 50 \tag{17}$$

```
> eval(subs(%,Demand));
```

$$P = 40 \tag{18}$$

```
> PriceCeiling := P = min(35,subs(Demand,P));
```

$$PriceCeiling := P = \min\left(35,\ 60 - \frac{2}{5}\ Q\right) \tag{19}$$

```
> eval(subs(%,Revenue));
```

$$R = \min\left(35,\ 60 - \frac{2}{5}\ Q\right)\ Q \tag{20}$$

```
> MarginalRevenue_Ceiling := MR = diff(rhs(%),Q):
> plot(rhs(MarginalRevenue_Ceiling),Q=0..75, title = "Marginal
  Revenue, Price Ceiling = 35");
```




```
> eval(subs(Revenue,PriceCeiling ,Production,LaborSupply,Profit));
```

$$\pi = \min\left(35,\ 60 - \frac{2}{5}\ \sqrt{10\,E - 5000}\right)\ \sqrt{10\,E - 5000}\ - 2\,E \tag{21}$$

```
> plot(rhs(%),E=600..1200);
```




```
> eval(subs(MarginalRevenueProduct,LaborSupply,ProfitMax));
```

$$MR\ MPL = 2 \tag{22}$$

```
> eval(subs(MarginalRevenue_Ceiling,Production,MarginalProduct,%))
  :
> plot([lhs(%),rhs(%)],E=800..1000,thickness=3,title="MRPL=w");
```




```
> [E=890,MRPL=evalf(subs
  (MarginalRevenueProduct,MarginalRevenue_Ceiling,Production,
  MarginalProduct,E=890,lhs(ProfitMax)))];
```

$$[E = 890, MRPL = 2.802242691] \quad \textbf{(23)}$$

```
> [E=891,MRPL=evalf(subs(MarginalRevenueProduct,
  MarginalRevenue_Ceiling,Production,MarginalProduct,E=891,lhs
  (ProfitMax)))];
```

$$[E = 891, MRPL = 0.7976976580] \quad \textbf{(24)}$$

```
> Cost := C = w*E;
```

$$Cost := C = w\,E \quad \textbf{(25)}$$

```
> eval(subs(solve(Production,{E}),Cost));
```

$$C = w\left(500 + \frac{1}{10}\,Q^2\right) \quad \textbf{(26)}$$

```
> MarginalCost := MC = diff(rhs(%),Q);
```

$$MarginalCost := MC = \frac{1}{5}\,w\,Q \quad \textbf{(27)}$$

```
>
```

```
> plot([rhs(MarginalRevenue_Ceiling),subs(LaborSupply,rhs
  (MarginalCost))],Q=0..100, title = "Marginal Revenue and
  Marginal Cost, Price Ceiling = 35");
```




```
> eval(subs(P=35,Demand));
```

$$35 = 60 - \frac{2}{5} Q \tag{28}$$

```
> Sol[Q,Ceiling] := solve(%,{Q});
```

$$Sol_{Q,\ Ceiling} := \left\{ Q = \frac{125}{2} \right\} \tag{29}$$

```
> evalf(%);
```

$$\{ Q = 62.50000000 \} \tag{30}$$

```
> eval(subs(Sol[Q,Ceiling],Production));
```

$$\frac{125}{2} = \sqrt{10\,E - 5000} \tag{31}$$

```
> Sol[E,Ceiling] := solve(%,{E});
```

$$Sol_{E,\ Ceiling} := \left\{ E = \frac{7125}{8} \right\} \tag{32}$$

```
> evalf(%);
```

$$\{E = 890.6250000\} \tag{33}$$

```
> eval(subs(LaborSupply,Revenue,PriceCeiling,Sol[Q,Ceiling],Sol[E,
  Ceiling],Profit));
```

$$\pi = \frac{1625}{4} \tag{34}$$

```
> evalf(%);
```

$$\pi = 406.2500000 \tag{35}$$

```
>
```